

```

PROGRAM GAUSSJ0
! plik: gj0_2.f90
! rozwiazuje uklad rownan liniowych postaci

!   a_11*x_1 + a_12*x_2 +...+a_1n*x_n = a_1,n+1, ... , a_1m
!   .....
!   a_n1*x_1 + a_n2*x_2 +...+a_nn*x_n = a_n,n+1, ... , a_nm

! n = liczba niewiadomych
! m = n + liczba kolumn wyrazow wolnych, m>=n+1

! metoda Gaussa-Jordana, bez porzadkowania kolumn, wersja elementarna
! wykorzystuje operacje na macierzach wg F90

! kodowal: M.Kuczma, 01.10.01; ostatnie zmiany: 6.10.07 (MSK)

IMPLICIT NONE
CHARACTER (LEN=80):: opis
INTEGER          :: i,j,k,n,m,il
REAL            :: c,d
REAL, ALLOCATABLE :: a(:, :)

OPEN(11, FILE='uklr.d', STATUS='OLD', ACTION='READ') !dane
OPEN(12, FILE='uklr.w', STATUS='NEW', ACTION='WRITE') !wyniki

READ(11, '(A80)') opis ! nazwa zadania, identyfikacja
WRITE(12, '(A80)') opis ! nazwa zadania -> opis wynikow
READ(11, '(A80)') opis ! n      m
READ(11, *) n,m

ALLOCATE(a(n,m)) ! nadanie wymiarow i przydzielenie pamieci mac. a

READ(11, '(A80)') opis ! Podaj wierszami elementy mac. a
DO i=1,n
! wczytanie elementow mac. a
  READ(11, *) (a(i,j), j=1,m)
END DO
WRITE(12, *) '      DANE - mac. a '
DO i=1,n
  WRITE(12, '(5E16.6)') (a(i,j), j=1,m)
END DO

! komendy rozwiazania ukladu rownan
pta_i:DO i=1,n
  c=a(i,i)
  a(i,:)=a(i,:)/c
  il=i+1
pta_k: DO k=1,n
  d=a(k,i)
  IF(k==i) CYCLE
  a(k,il:m)=a(k,il:m)-d*a(i,il:m)
END DO pta_k
END DO pta_i
! koniec rozwiazywania ukladu rownan

! wydruk rozwiazania x_i
WRITE(12, *) '      Wyniki x_i wg gj0_2.f90'
DO k=n+1,m
  j=k-n
  WRITE(12, *) 'dla kolumny',j
  DO i=1,n
    WRITE(12, *) i, a(i,k)
  END DO
END DO
STOP
END PROGRAM GAUSSJ0

```